Laser Beveling in polyFORTH

Thomas R. Anthony

General Electric Research and Development Center
Schenectady, NY 12301

Abstract

High power thyristors capable of carrying 1000 amperes and of blocking up to 6500 volts, used for high voltage DC transmission applications, must have their edges beveled to have the lowest possible electric fields at their surfaces to prevent premature surface breakdown in reverse bias. A computer-controlled system has been developed for the General Electric High Voltage DC Transmission Operation in Collingdale, Pa., to produce bevels on the edges of their high-power thyristors. The system uses an IBM PC connected to three micromanipulators and a Q-switched laser. Various laser-cutting algorithms coded in polyFORTH* have been developed for beveling. Each algorithm requires different processing times and produces different quality bevels in terms of thyristor surface breakdown resistance.

Introduction

When a P-N junction is put in reverse bias, the internal electric field near the junction forces mobile charges of opposite polarity to separate from a zone on both sides of the junction called the depletion zone. Under these conditions, the absence of mobile charges in the depletion zone causes the reverse-biased junction to be nonconducting. Since a forward-biased P-N junction has a relatively high conductivity, the P-N junction is said to be a rectifier. Although the depletion zone is devoid of mobile charges such as electrons and holes, ionized immobile atoms are present. The charges of these immobile ions are opposite on opposing sides of the P-N junction. To satisfy charge neutrality, the total charge of these immobile ions on opposing sides of the P-N junction in the depletion zone must be equal.

In the bulk of a semiconductor crystal with a planar P-N junction, the charge distribution of these immobile ions is directly proportional to the respective impurity concentrations on opposing sides of the P-N junction. Thus, in the bulk of the crystal, the width of the depletion zone and the peak electric field in the crystal are fixed solely by the existing impurity concentration profiles. Near the intersection of the P-N junction and the external surface of the crystal, however, the total charge of immobile ions on opposing sides of the P-N junction is determined not only by the impurity concentration profiles in the crystal but also by the geometric profile of the surface of the crystal. That is, the local charge distribution near the surface on either side of the P-N junction can be increased or decreased by the physical addition or removal, respectively, of crystalline material.

Such surface contouring can reduce the peak electric field at the surface in the following manner. Figure 1 shows a cross-sectional view of a beveled thyristor in reverse bias. The effect of the beveling is to remove material from the lightly doped N-type region and to add material to the heavily doped P-type region in the near-surface regions around the blocking...

*polyFORTH is a registered trademark of FORTH, Inc., Hermosa Beach, CA.
junction. To maintain overall charge neutrality of immobile ions in the depletion region, the depletion region in the near-surface region must expand outwards into the lightly doped N-type region to encompass enough additional negative ions to compensate for those removed by beveling. Conversely, in the heavily doped P-type region, the depletion zone in the near-surface region must contract towards the junction to reduce the number of positive ions effectively added by beveling to the depletion zone.

![Diagram of a thyristor with beveled edges showing the depletion zone profile](image)

**Figure 1.** A reverse-biased thyristor with beveled edges showing the depletion zone profile in the near-surface and bulk regions around the blocking junction.

The resulting boundaries of the depletion zone are shown in Fig. 1 at the blocking junction. Because of the difference in impurity concentration between the lightly doped N and heavily doped P regions, the near-surface expansion of the depletion zone in the N-type region exceeds the near-surface contraction in the P-type region. Consequently, a net increase in the depletion zone width exists in the near-surface region. The increased width of the depletion zone in the near-surface region decreases the peak electric field strength at the surface relative to the electric field strength in the crystal bulk, since the same voltage drop occurs across all portions of the depletion zone. Since the electric breakdown strength at surfaces is lower than in the bulk because of dangling atomic bonds and mobile adsorbed impurities at surfaces, a reduction in the surface electric field by surface contouring can significantly increase the overall breakdown voltage of a device.

With a typical 45° bevel angle, the surface electric field is less than half of the bulk electric field.\[^{[1]}\] This decrease in the surface electric field strength is enough to insure that breakdown will occur first in the bulk where ideal plane junction breakdown voltages are achieved.

Beveling of thyristor edges is currently accomplished by directing an airstream containing SiC grit onto the edge of a rotating thyristor at an appropriate angle. This grit blasting is followed by a light edge etch in acid to remove a damaged surface layer. Although beveling by grit blasting is an accepted industrial method, two problems associated with grit blasting have instigated investigations into finding a new method to replace it. These two problems are: 1) the high cost of SiC powder; 2) the general housecleaning problem of using
a fine abrasive dust in a semiconductor manufacturing facility.

A potential replacement for the grit blasting method is a process utilizing laser cutting. A laser beveler is attractive in that it eliminates the material cost of the SiC powder and makes the beveling procedure a much cleaner process.

**Comparison of Beveling Algorithms**

Four basic algorithms have been defined for laser beveling a thyristor, namely DOWN-CUT, MULTIPLE-DOWN-CUT, TRIANGLE-CUT and WEDGE-CUT. Schematic cross-sectional diagrams of these laser beveling algorithms are shown in Figs 2, 3, 4 and 5, respectively. The wafer is continuously rotated about an axis perpendicular to the surface, and passing through the center of the wafer during laser cutting. Each small block on these diagrams shows the material removed by the laser during one wafer rotation. The arrows represent the movement of the laser at the end of each wafer rotation. The summation of all these movements comprise the laser beveling algorithm programmed in polyFORTH. A comparison of these four algorithms should consider both the time required to form a bevel by each method and the quality of the resulting bevel.

The time \( \tau_{DC} \) required to form a bevel with DOWN-CUT (Fig. 2) is equal to the time per wafer revolution \( \frac{2 \pi R}{V} \) times the number of cuts \( \frac{T}{D} \) required to cut through the wafer.

\[
\tau_{DC} = \frac{2 \pi T R}{D V}
\]  

(1)

where \( R \) is the radius of the wafer, \( V \) is the linear velocity at the edge of the wafer, \( T \) is the vertical distance through the wafer and \( D \) is the depth of material removed by each pass of the laser.

The time \( \tau_{MDC} \) needed to bevel a wafer with MULTIPLE-DOWN-CUT (Fig. 3) is the number \( N \) of DOWN-CUTs (Fig. 2) used times the time per DOWN-CUT given by Eq 1.

\[
\tau_{MDC} = \frac{2 \pi N T R}{D V}
\]  

(2)

The time \( \tau_{TC} \) required to bevel a wafer with the TRIANGLE-CUT algorithm (Fig. 4) may be calculated as follows. The number of wafer revolutions or equivalent laser passes needed to form the triangular cut shown in cross section in Figure 4 is equal to the cross-sectional area of the cut \( \frac{T^2}{2} \) divided by the area per cut DW removed by each laser pass, where \( W \) is the width of each laser pass. The total time to produce the triangular cut is the product of this number of wafer revolutions and the time per wafer revolution.

\[
\tau_{TC} = \frac{\pi R T^2}{D V W}
\]  

(3)

The time \( \tau_{WC} \) that it takes to form a bevel with the WEDGE-CUT algorithm shown in Figure 5 can be similarly calculated. The cross-sectional area of the wedge cut is \( \frac{W T^2}{D} \) and the area removed per laser pass is DW. Consequently, the time to produce the wedge cut is

\[
\tau_{WC} = \frac{\pi R T^2}{V D^2}
\]  

(4)

Equations 1–4 can be used to estimate the times required to form bevels with the four different cutting algorithms by substituting reasonable values for the variables in these expressions. The radius \( R \) for the thyristor of interest is 3.8 cm (1.5 inches). The vertical thickness \( T \) through this same thyristor at an angle of 45 degrees to the vertical is 0.175 cm (70 mils). The cut depth \( D \) per laser pass is about 0.0125 cm (5 mils). The maximum velocity \( V \) of the edge of the wafer is related to the repetition rate and beam diameter of the laser. With Q-switched
Figure 2. A schematic cross-sectional diagram of a laser cutting a circumferential cut through a rotating wafer at a 45° angle to the laser beam with sequential passes (DOWN-CUT).
Figure 3. A schematic cross-sectional diagram of a laser cutting multiple circumferential cuts through a rotating wafer at a 45° angle to the laser beam (MULTIPLE-DOWN-CUT).
Figure 4. A schematic cross-section of a laser cutting a triangle cut through a rotating wafer at a 45° angle to the laser beam to form an edge bevel (TRIANGLE-CUT).
Figure 5. A schematic cross-section of a laser cutting a wedge cut through a rotating wafer at a 45° angle to the laser beam to form an edge bevel (WEDGE-CUT).
multipulse laser cutting, the laser forms the cut by overlapping a row of individual drill holes. The maximum speed is attainable when the individual drill holes just touch. With a laser pulse repetition rate of 4 kHz and a beam diameter of 0.0025 cm, the maximum velocity \( V \) of the edge of the rotating wafer is 10 cm/sec. If the number \( N \) of down cuts in the MULTIPLE-DOWN-CUT algorithm is taken as ten, the times required to bevel the thyristor by the four cutting algorithms are given in Table I.

<table>
<thead>
<tr>
<th>Algorithm</th>
<th>Time (minutes)</th>
</tr>
</thead>
<tbody>
<tr>
<td>DOWN-CUT</td>
<td>0.56</td>
</tr>
<tr>
<td>MULTIPLE-DOWN-CUT</td>
<td>5.6</td>
</tr>
<tr>
<td>WEDGE-CUT</td>
<td>4</td>
</tr>
<tr>
<td>TRIANGLE-CUT</td>
<td>20</td>
</tr>
</tbody>
</table>

By using a larger cut depth and/or beam diameter, these times can probably be reduced by a factor of two. However, a factor of ten reduction would require a more powerful laser that might result in excessive damage to the thyristor during beveling.

These beveling times have been calculated for 3-inch diameter high-voltage thyristors. Beveling times for smaller thyristors will be reduced by the square of the ratio of their diameters for the DOWN-CUT and MULTIPLE-DOWN-CUT algorithms and by the cube of the ratio of their diameters for the WEDGE-CUT and TRIANGLE-CUT algorithms. Table II gives the beveling times for these four algorithms for a 1.5-inch diameter thyristor.

<table>
<thead>
<tr>
<th>Algorithm</th>
<th>Time (minutes)</th>
</tr>
</thead>
<tbody>
<tr>
<td>DOWN-CUT</td>
<td>0.14</td>
</tr>
<tr>
<td>MULTIPLE-DOWN-CUT</td>
<td>1.4</td>
</tr>
<tr>
<td>WEDGE-CUT</td>
<td>0.5</td>
</tr>
<tr>
<td>TRIANGLE-CUT</td>
<td>2.5</td>
</tr>
</tbody>
</table>

The quality of the bevels for the large thick thyristors generally is an inverse function of the beveling times shown in Tables I and II. The simple one-slice bevel of the DOWN-CUT algorithm usually has a considerable amount of drilling debris on it and sometimes fails to go completely through the wafer in all positions. In contrast, the wide-open TRIANGLE-CUT always goes through the wafer and is practically free of drilling debris. Figure 6 shows
a high-quality bevel produced on a 76 mm thyristor with the TRIANGLE-CUT algorithm. MULTIPLE-DOWN-CUT and WEDGE-CUT produce bevels of intermediate quality.

For smaller and thinner thyristors, the simple one-slice bevel of DOWN-CUT may give high enough quality bevels so that the more elaborate and longer cutting algorithms are not necessary.

**Laser Beveling Instrumentation**

The first step in building a laser beveling system is to select a laser type. The three general choices are a continuous wave (cw) laser, a pulsed laser and a Q-switched laser. Early experiments showed that the cw and pulsed lasers were unsuitable because they generated cracks and stresses that extended more than 1mm into the bulk of the wafer from the cut surface. In contrast, laser cutting with a Q-switched laser produced a depth of thermal damage that was removable in a subsequent etching step. In experiments with the Q-

Figure 6. Beveled edge of a 76 mm diameter thyristor mounted on a tungsten backing plate. The laser beveling algorithm was the TRIANGLE-CUT.
switched laser, the laser pulse length was 200 nsec. Hence, the thermal diffusion distance equals 12 microns. A subsequent edge etch was able to remove this damage. Good breakdown voltages of high-power thyristors beveled manually with this Q-switched laser have been achieved.

The laser system selected for laser beveling was a Model 117 Nd:YAG laser built by Quantronix Corp. of Smithtown, New York. Its maximum output power in a TEM 00 output mode is about 18 watts cw power. The laser was operated at a frequency of 4 kHz. This frequency represents a compromise between achieving the highest peak and highest average output powers of this laser system.

Focal plane movements and wafer positioning were accomplished with an X-Y Positioning Table Model DC-33 obtained from Design Components Inc. of Medfield, Mass. Two synchronous stepping motors (Type MO63-FD06), capable of two hundred steps per revolution at 100 ounce-inch torque, were obtained from Superior Electric of Bristol, Conn. Computer commands from the IBM PC to these stepping motors were used to position the X-Y table. A Rotary Table Model RT 6180, made by Design Components Inc., was attached at a predetermined angle to the X-Z Table to rotate the semiconductor wafer that was being beveled. The rotary table was driven by a Bodine Series 200 Control Motor type NSH-12R. A stepping motor was not used to drive the rotary table to avoid any synchronization between the wafer rotation and the laser pulsing. This lack of synchronization produces a smoother cut surface than would be the case with synchronization.

The computer chosen for the laser beveling system was an IBM PC with 64K of memory and an I/O interface board with three I/O ports.

**polyFORTH Coding**

A version of FORTH called polyFORTH II, Version 1.1, for the IBM PC obtained from FORTH, Inc. of Hermosa Beach, Calif. was used as the programming language of the laser beveling system.

```
Screen #1
0 ( LASER BEVELEER SCREEN #1) PAGE HEX
1 210 CONSTANT ROTATE-PORT
2 207 CONSTANT XY-PORT
3 80 CONSTANT Initialization-CONSTANT
4 DECIMAL
5 2 CONSTANT DELAY
6 3 CONSTANT UP-ON
7 2 CONSTANT UP-OFF
8 1 CONSTANT DOWN-ON
9 0 CONSTANT DOWN-OFF
10 5 CONSTANT LEFTWARD-ON
11 4 CONSTANT LEFTWARD-OFF
12 7 CONSTANT RIGHTWARD-ON
13 6 CONSTANT RIGHTWARD-OFF
14 38 CONSTANT RADIUS
15 1 CONSTANT WIDTH-OF-BEAM
```

In Screen #1, a number of constants are defined and given values. ROTATE-PORT is the I/O port where signals are sent to cause the thyristor to rotate, and XY-PORT is the I/O port where signals from the computer are sent to cause the thyristor to move in the xy-plane.
The **INITIALIZATION-CONSTANT** is used to initialize the 8255 Intel Programmable Parallel Interface Controller on the I/O board that is used for the XY and **ROTATE I/O** ports. Parallel transmission is useful for high-speed applications using devices which are not too far away from the computer. Since no special timing is involved in parallel transmission, bytes may be sent to parallel I/O ports as fast as or as slow as software dictates. This feature is important in controlling the speed of the XY stepping motors connected to the 8255 I/O ports. Outputting the **INITIALIZATION-CONSTANT** configures the 8255 into an output transmission mode for all 24 I/O lines hooked to the 8255. Eight of these output lines serve as the **ROTATE I/O** port while a second set of eight output lines service the XY I/O port. The **INITIALIZATION-CONSTANT** is sent to **XY-PORT** when the system is started.

The constant **DELAY** is used to match the inertial dynamics of the stepping motors by inserting a delay time between output pulses from the computer.

Eight direction constants are defined and given suitable values. When **UP-ON** is sent to the **XY-PORT**, the XY-stage will start moving the thyristor upwards. The thyristor will continue to move up until the constant **UP-OFF** is sent to the XY-PORT which stops the upward motion. Similar start and stop constants (**DOWN-ON**, **DOWN-OFF**, **LEFTWARD-ON**, **LEFTWARD-OFF**, **RIGHTWARD-ON**, **RIGHTWARD-OFF**) are defined for the other three principal directions of motion of the system. These constants are also sent to the **XY-PORT** to cause the thyristor to start and stop moving in the respective directions.

The constant **RADIUS** is the radius in mm of the thyristor and is used later in converting rotational speeds to linear cutting rates in some of the laser cutting algorithms.

The constant **WIDTH-OF-BEAM** is the width of the laser beam in mils and is used in one of the laser cutting algorithms.

Screen #2

0 DECIMAL ( LASER BEVELER SCREEN #2 )
1 ( INITIALIZE ) INITIALIZATION--CONSTANT XY-PORT OUTPUT
2 : START INITIALIZATION-CONSTANT XY-PORT OUTPUT ;
3 : DIRECTION ( -- DIRECTION-OFF DIRECTION-ON )
4 CREATE C, C, DOES> DUP C@ SWAP 1+ C@ ;
5
6 UP-ON UP-OFF DIRECTION UPWARDS
7 DOWN-ON DOWN-OFF DIRECTION DOWNWARDS
8 RIGHTWARD-ON RIGHTWARD-OFF DIRECTION RIGHTWARDS
9 LEFTWARD-ON LEFTWARD-OFF DIRECTION LEFTWARDS
10
11 : ONE-STEP ( DIRECTION -- )
12 XY-PORT OUTPUT DELAY MS XY-PORT OUTPUT ;
13
14
15

The stepping motors used in the laser beveler system turn one step when their control boards receive a positive pulse. **ONE-STEP** causes a stepping motor to turn exactly one step by directing the control board to output a positive pulse for 2 milliseconds ( **DELAY MS** ) and then pulling the voltage level back to ground to prepare for the next positive pulse. By changing the value of the bytes output to the XY-PORT, different I/O lines, and thus different stepping motors, are affected. For example, **UPWARDS** puts two bytes on the parameter stack that **ONE-STEP** outputs to the XY-PORT. This causes the stepping motor controlling vertical motion to turn one step in the upwards direction.
Screen #3

0 DECIMAL ( LASER BEVELER SCREEN #3 )
1 : MOTION-EQUALS ( DIRECTION — ) CREATE , , DOES>
2 SWAP 2* 0 DO DUP DUP 2+ @ SWAP @ ONE-STEP LOOP DROP ;
3 UPWARDS MOTION-EQUALS UP ( DISTANCE — )
4 DOWNWARDS MOTION-EQUALS RIGHT ( DISTANCE — )
5 RIGHTWARDS MOTION-EQUALS RIGHT ( DISTANCE — )
6 LEFTWARDS MOTION-EQUALS LEFT ( DISTANCE — )
7 : U UP ;
8 : D DOWN ; : DN DOWN ;
9 : R RIGHT ; : RT RIGHT ;
10 : L LEFT ; : LF LEFT ;
11 : UP-LASER DOWN ;
12 : DOWN-LASER UP ;
13 : RIGHT-LASER LEFT ;
14 : LEFT-LASER RIGHT ;
15

The motion commands, UP, DOWN, LEFT and RIGHT, defined in Screen #3, refer to motions of the thyristor that is being beveled. In programming various beveling algorithms, it was found that it was easier to think about these algorithms in terms of motions of the cutter (laser) rather than the object (thyristor) being cut. Consequently, the equivalent cutter commands, namely, UP-LASER, DOWN-LASER, RIGHT-LASER and LEFT-LASER, were also defined.

Screen #4

0 DECIMAL ( LASER BEVELER SCREEN #4)
1 ( CONVERT MILLIREVOLUTIONS/SEC TO OUTPUT BYTE AT HIGH SETTING)
2 ( MILLIREV/SEC RANGE FROM 0 —> 53 AT HIGH MOTOR SETTING)
3 CREATE MILLIREV/SEC->OUTPUT-HIGH 0 C, 15 C, 17 C, 19 C, 21 C,
4 24 C, 26 C, 28 C, 30 C, 33 C, 36 C, 38 C, 41 C, 44 C, 46 C,
6 77 C, 80 C, 84 C, 87 C, 90 C, 93 C, 96 C, 100 C, 102 C, 105 C,
7 108 C, 112 C, 115 C, 118 C, 121 C, 125 C, 128 C, 132 C, 136 C,
8 140 C, 145 C, 150 C, 155 C, 160 C, 166 C, 172 C, 178 C, 186 C,
9 195 C,
10
11
12
13
14
15
Screen #5
0 DECIMAL (LASER BEVELER SCREEN #5)
1 (CONVERT MILLIREvolUTIONS/SEC TO OUTPUT BYTE AT LOW SETTING)
2 (MILLIREV/SEC RANGE FROM 0 —> 45 AT LOW MOTOR SETTING)
3 CREATE MILLIREV/SEC—>OUTPUT-LOW 0 C, 17 C, 20 C, 22 C, 25 C,
4 28 C, 31 C, 34 C, 37 C, 40 C, 43 C, 46 C, 49 C, 53 C, 56 C,
5 59 C, 62 C, 66 C, 69 C, 73 C, 76 C, 80 C, 84 C, 88 C, 92 C,
6 97 C, 101 C, 106 C, 110 C, 116 C, 120 C, 125 C, 130 C, 135 C,
7 140 C, 144 C, 150 C, 155 C, 160 C, 166 C, 171 C, 176 C, 182 C,
8 188 C, 195 C, 202 C,
9
10
11
12
13
14
15

Screens 4 and 5 provide lookup tables in order to convert millirevolutions per second into a byte to be sent to the ROTATE-PORT. This will cause the thyristor to rotate at a given angular velocity when a manual switch on the control chassis is set to a high or low setting. The lookup table was required because a nonlinear relationship exists between the revolution rate and the byte value sent to the ROTATE-PORT (Figures 7a and 7b).

Screen #6
0 DECIMAL (LASER BEVELER SCREEN #6)
1 VARIABLE MILLIREV/SEC
2 : MM/SEC (VELOCITY —) 100 * 1000 628 RADIUS * */ MILLIREV/SEC ! ;
3 : CM/SEC (VELOCITY —) 10 * MM/SEC ;
4 : 1/10-SEC 100 MS ;
5 : 1/4-REV [ 1000 10 * 4 / ] LITERAL
6 MILLIREV/SEC @ / 0 DO 1/10-SEC LOOP ;
7 : WAIT-ONE-REVOLUTION 4 0 DO 1/4-REV LOOP ;
8 : IREV WAIT-ONE-REVOLUTION ;
9 : ROTATE-LOW MILLIREV/SEC @ DUP 45 >
10 "ABORT" VELOCITY OUT OF RANGE OF 0-11 MM/SEC. REDO, "
11 MILLIREV/SEC—>OUTPUT-LOW + @ ROTATE-PORT OUPUT ;
12 : ROTATE-HIGH MILLIREV/SEC @ DUP 53 >
13 "ABORT" VELOCITY OUT OF RANGE OF 0-13 MM/SEC. REDO, "
14 MILLIREV/SEC—>OUTPUT-HIGH + @ ROTATE-PORT OUPUT ;
15 : ROTATE (ROTATE-HIGH IS DEFAULT) ROTATE-HIGH ;

MM/SEC takes a linear velocity value and converts it to an equivalent angular velocity in millirevolutions per second by taking 1000 times 2 times \( \pi \) times the RADIUS divided by the linear velocity in mm/sec. This angular velocity is then stored in the variable MILLIREV/SEC.

1/10-SEC forces the computer to pause for 1/10 of a second by using the polyFORTH primitive MS. The word 1/4-REV waits for a time period required for the thyristor platform to turn 1/4 of a revolution at its current rotational velocity. This definition is used later to produce four symmetrical radial cuts in a thyristor. WAIT-ONE-REVOLUTION does what its name implies.
Figure 7a. A graph of the rotation rate of the thyristor platform in millirevolutions/sec versus the byte value sent to the ROTATE-PORT I/O port for the high motor-speed switch setting.

ROTATE-LOW causes the thyristor platform to rotate with the manual motor setting on "low". If the desired rotation rate is out of the range that the motor is capable of at the low setting, ABORT" prints out the message VELOCITY OUT OF RANGE OF 0-11 MM/SEC. REDO. If the desired rotation is in the acceptable range, the rotation rate is converted by the words MILLIREV/SEC→OUTPUT-LOW + @ to an appropriate byte to be sent to the I/O port of the motor control. Similarly, ROTATE-HIGH causes the thyristor platform to rotate with the manual motor setting on "high".
Figure 7b. A graph of the rotation rate of the thyristor platform in millirevolutions/sec versus the byte value sent to the ROTATE-PORT I/O port for the low motor-speed switch setting.

Screen #7

0 DECIMAL (LASER BEVELER SCREEN #7)
1 : HALT-ROTATION 0 MM/SEC ROTATE ;
2 : HALT HALT-ROTATION ;
Screen #8
0 DECIMAL ( LASER BEVELER SCREEN #8)
1 : DOWN-CUT ( CUT-DEPTH TOTAL-THICKNESS —) 2DUP
2  0 DO DUP DOWN-LASER WAIT-ONE-REVOLUTION DUP +LOOP DROP
3  0 DO DUP UP-LASER DUP +LOOP DROP ;
4 : MULTIPLE-DOWN-CUT ( CUT-DEPTH TOTAL-THICKNESS NUMBER —)
5  0 DO 2DUP DOWN-CUT 1 UP-LASER 1 RIGHT-LASER LOOP 2DROP ;
6 : RADIAL-CUT ( CUT LENGTH —)
7  ( USES 5 MIL DOWN-CUT and 100 MIL TOTAL-THICKNESS)
8 HWT-ROTATION
9  0 DO
10  5 100 2DUP 0 DO DUP DOWN-LASER 2 MS DUP +LOOP DROP
11  0 DO DUP UP-LASER DUP +LOOP DROP
12  1 UP-LASER 1 RIGHT-LASER
13 LOOP ;
14 : RADIAL-CUT-RETURN ( CUT-LENGTH —)
15  0 DO 1 DOWN-LASER 1 LEFT-LASER LOOP ;

DOWN-CUT causes the laser to make a series of circumferential cuts of a depth CUT-DEPTH one on top of another through a thickness TOTAL-THICKNESS. A schematic cross-sectional diagram of this cutting algorithm is shown in Figure 2. A laser beam is removing material in multiple steps from a rotating wafer whose axis of rotation and surface normal is at an angle of 45° to the laser beam. By cutting all of the way through the wafer, a 45° bevel is formed at the new edge of the wafer.

The cutting action of MULTIPLE-DOWN-CUT is shown schematically in Fig. 3. MULTIPLE-DOWN-CUT makes a series of contiguous DOWN-CUTS moving 1 mil to the right after completion of each DOWN-CUT. Because the wafer is at a 45 degree angle to the laser beam, the laser must be moved up 1 mil for each 1 mil it is moved to the right. Thus, MULTIPLE-DOWN-CUT consists essentially of three statements, DOWN-CUT, UP-LASER and RIGHT-LASER.

RADIAL-CUT, causes the laser to make a radial cut in the wafer from its outer periphery inward a distance equal to CUT-LENGTH. The cut is carried out while the wafer is at a 45 degree angle to the laser beam. Figure 8 shows a schematic diagram of this cutting process. RADIAL-CUT is a multiple laser pulse method where twenty cuts that are each 5 mils in depth are used to cut through the wafer at each position. The laser is moved 1 mil to the right and 1 mil upwards after each through-cut until a radial cut with a length of CUT-LENGTH is made.

RADIAL-CUT begins by first stopping any wafer rotation with the word HALT-ROTATION. The first inner DO LOOP of RADIAL-CUT produces the through-cut at each laser position. A 2 millisecond delay between each of the twenty laser pulses is used to avoid a mechanical resonance in the system. The second inner DO LOOP retracts the laser to its initial position after the through-cut at each position is completed. The outer DO LOOP iterates the through-cut at sequential laser positions spaced 1 mil apart until the cut length equals CUT-LENGTH.

RADIAL-CUT-RETURN is an operation that returns the laser to the outer periphery of the wafer following a radial cut, in order to prepare for another radial cut after the wafer is rotated.
Figure 8. A schematic cross-sectional diagram of a laser cutting a radial cut through a wafer at a 45° angle to the laser beam (RADIAL-CUT).
Screen #9
0 DECIMAL ( LASER BEVELER SCREEN #9)
1 : TURN 6 MM/SEC ROTATE ;
2 : CUT-R ( CUT-LENGTH — ) CR DUP RADIAL-CUT RADIAL-CUT-RETURN ;
3 : RADIAL-QUARTER-CUTS ( CUT-LENGTH — )
4     HALT-ROTATION CR
5     DUP ." 1ST CUT" CUT-R
6     TURN 1/4-REV
7     DUP ." 2ND CUT" CUT-R
8     TURN 1/4-REV
9     DUP ." 3RD CUT" CUT-R
10    TURN 1/4-REV
11    ." 4TH CUT" CUT-R ;
12
13
14
15

TURN makes the thyristor platform rotate such that the linear speed at the periphery of the thyristor will be 6 mm/sec. This command is used in a cutting algorithm to make multiple radial cuts in the thyristor. The speed of 6 mm/sec was chosen as the fastest speed at which the thyristor platform could be stopped without appreciable angular overshoot.

RADIAL-QUARTER-CUTS forms four symmetrically placed radial cuts of length CUT-LENGTH around the periphery of a thyristor. These radial cuts are used to break up a ring of silicon that is produced when a circumferential through-cut is made around the thyristor. This ring falls back on the thyristor when a circumferential cut is completed and may interfere with laser cuts that follow. By cutting the ring into quarter sections, pieces of the ring can fall away freely to the ground, leaving the thyristor completely unobstructed.

Screen #10
0 DECIMAL ( LASER BEVELER SCREEN #10)
1 : TRIANGLE-CUT ( CUT-DEPTH TOTAL-THICKNESS — )
2     OVER + OVER
3     DO
4     1 0 DO 2 RIGHT-LASER WAIT-ONE-REVOLUTION 2 +LOOP
5     DUP 1 + LEFT-LASER
6     DUP DOWN-LASER DUP
7     +LOOP DROP ;
8
9
10
11
12
13
14
15

TRIANGLE-CUT forms a right triangle cut through a rotating wafer whose normal is at an angle of 45 degrees to the impinging laser beam. Figure 4 shows a schematic cross section of this cutting algorithm. TRIANGLE-CUT removes sequential layers with the laser
gradually stepping from left to right on each layer as the wafer rotates. Thus, the last cut completed on any layer is at the final bevel face. This cutting sequence helps to keep the bevel face as clean from cutting debris as possible. After each layer is cut, the laser is moved back to the wafer surface on the left and the focal plane of the laser is lowered by the CUT-DEPTH. Then a new layer is removed.

Screen #11
0 DECIMAL ( LASER BEVELER SCREEN #11)
1 : WEDGE-CUT ( CUT-DEPTH TOTAL-THICKNESS —)
2 CR CR OVER / 0 SWAP
3 DO
4 CR " CUTTING LEVELS REMAINING = " I 1+ . CR
5 I LEFT-LASER
6 DUP DOWN-LASER
7 I 0 DO WIDTH-OF-BEAM RIGHT-LASER
8 WAIT-ONE-REVOLUTION WIDTH-OF-BEAM
9 +LOOP
10 -1
11 +LOOP DROP CR " WEDGE-CUT COMPLETE" CR ;

WEDGE-CUT forms a bevel on a thyristor by cutting a wedge cross-sectional cut through a rotating wafer as illustrated in Figure 5. The wedge angle in radians in this algorithm is equal to the number of cuts required to pass through the wafer divided by the vertical thickness in mils of the wafer. For example, with a CUT-DEPTH equal to 5 mils and a vertical TOTAL-THICKNESS of 100 mils, the number of cuts required is 20. Thus, the wedge angle in radians is 20/100 = 0.2.

As shown in Figure 5, WEDGE-CUT works by stepping and cutting across sequential layers from left to right. After each layer is cut, the length of the return sweep to the left is shortened from the length of the return sweep of the previous layer. The words I LEFT-LASER produce the ever-shortening return sweep, while the inner DO LOOP containing RIGHT-LASER does the cutting job on each layer. DOWN-LASER lowers the focal plane of the laser after the removal of each layer is completed. At each cutting level, a message is sent to the terminal to tell the operator how many layers remain to be cut.

Acknowledgements

Laser beveling of thyristors is an idea that has been supported by Dante Picone of General Electric's High Voltage Power Thyristor Operation in Collingdale, PA. Without his ideas, support, and thyristors, this work would not have been possible.

Implementation of this system was facilitated by my skilled colleagues at the Research and Development Center. Sig Ludke designed the 8255 interface board and the motor controllers. Bob Ettinger designed the mechanical portions of this system. Thanks are also due to John Johnson who introduced me to FORTH in his excellent course.

References
Manuscript received January 1984.

Dr. Anthony received a BS degree in Physics from the University of Florida in 1962 and MS and PhD degrees in Applied Physics from Harvard University in 1964 and 1967, respectively. He joined the General Electric Research & Development Center in 1967 where he has worked in the areas of diffusion, radiation damage, thermomigration, electromigration, amorphous metals and semiconductor and laser material processing. In 1978, he was awarded a Coolidge Fellowship for his work on thermomigration and semiconductor processing.